

This Maple file accompanies the paper "The Tu-Deng Conjecture holds almost surely" from Lukas Spiegelhofer and Michael Wallner.

```
#####
#####
```

Abstract:

The Tu-Deng Conjecture is concerned with the sum of digits of n in base 2 (the Hamming weight $w(n)$ of the binary expansion of n) and states the following: assume that k is a positive integer and $0 < t < 2^k - 1$. Then

$$|\{ (a,b) \text{ in } \{0, \dots, 2^k - 2\}^2 : a+b = t \text{ mod } 2^k - 1, w(a)+w(b) < k \}| < 2^{k-1} + 1.$$

We prove that the Tu-Deng conjecture holds almost surely in the following sense: The proportion of t in $[1, 2^k - 2]$ is such that the above inequality holds approaches one as k tends to infinity.

Moreover, we prove that the Tu-Deng conjecture implies a conjecture due to T. W. Cusick concerning the sum of digits of n and $n+t$.

```
#####
#####
```

In this worksheet we compute the asymptotic expansions of certain generating function associated with the expected value and the variance of a random variable. These results are then used to prove the asymptotic version of the Tu-Deng Conjecture by showing concentration of the random variable with Chebyshev's inequality.

For the computations the gfun package needs to be loaded (mostly for the gdev command). For the latest version see Salvy's homepage <http://perso.ens-lyon.fr/bruno.salvy/software/the-gfun-package/>. For this computations gfun version 3.76 was used.

Additionally, we use the gdev command part of the gdev package and available e.g., in the algolib package: <http://algo.inria.fr/libraries/>.

(Note that if you load the algolib package, the gfun package must be loaded first, because algolib contains a possibly older version of gfun.)

```
> restart;
#libname := "<path>\gfun.mla", "<path>\alolib.mla", libname;
with(gfun): gfun:-version;
libname := "D:\lib\gfun.mla", "D:\lib", "/home/michaelwallner/lib/gfun.mla",
"/home/michaelwallner/lib", "/usr/local/Maple2017/lib"
```

3.76

(1)

Proof of Proposition 4.2 - The asymptotics of the first moment

We want to compute $[x^n y^{(n-1)}] G(x,y)$

```
> G := (2 - 4*x*y - x - x*y^2) / ((1-y)*(1-2*x*y)*(1-x*(1+y)^2));
map(simplify, convert(gdev(convert(gdev(G, x, 6), polynom), y, 6),
```

polynom));
 add(coeff(coeff(% , x, n), y, n-1)*x^n*y^(n-1), n=1..5);

$$G := \frac{-xy^2 - 4xy - x + 2}{(1-y)(-2xy+1)(1-x(1+y)^2)}$$

$$x^5 + x^4 + x^3 + x^2 + x + 2 + (11x^5 + 9x^4 + 7x^3 + 5x^2 + 5x + 2)y + (56x^5 + 37x^4 + 22x^3 + 15x^2 + 6x + 2)y^2 + (176x^5 + 93x^4 + 50x^3 + 19x^2 + 6x + 2)y^3 + (386x^5 + 179x^4 + 65x^3 + 20x^2 + 6x + 2)y^4 + (670x^5 + 235x^4 + 71x^3 + 20x^2 + 6x + 2)y^5$$

$$386x^5y^4 + 93x^4y^3 + 22x^3y^2 + 5x^2y + x \tag{1.1}$$

First, we shift the function G by y in order to be able to compute the diagonal $[x^n y^n] G(x,y)$ and not $[x^n y^{(n-1)}] G(x,y)$

Then, we substitute one variable with t and the other one with x/t. The constant term of this function is the desired diagonal.

For more information on this method see Stanley's book "Enumerative Combinatorics Volume 2", Chapter 6.3.

This is the only choice for the substitution such that roots of denom are power series in x.

The leading coefficient of the denominator with respect to t is $(-1+x)$.

> **GT:=simplify(subs({x=t,y=x/t},y*G));**
den:=[solve(denom(%),t)];
map(gdev,den,x,2);

$$GT := - \frac{x(t^2 + (4x-2)t + x^2)}{(t-x)(2x-1)(t^2 + 2xt + x^2 - t)}$$

$$den := \left[x, -x + \frac{1}{2} + \frac{\sqrt{-4x+1}}{2}, -x + \frac{1}{2} - \frac{\sqrt{-4x+1}}{2} \right]$$

$$[x, 1 - 2x + O(x^2), x^2 + 2x^3 + O(x^4)] \tag{1.2}$$

We perform a partial fraction decomposition with respect to t.

The variables A,B, and C are constants in x but independent of t.

Note that we omitted the factor $1/(2*x-1)$ here.

Note that at the end we are interested in extracting the coefficient of the constant term wrt to t. This is the diagonal of the initial BGF.

Furthermore, our solution is a power series in x. Thus, in the following Ansatz only the second term with a B gives a function with nonnegative terms in t.

In the first one, in order to ensure a power series in x, we have only negative powers of t. The same holds for the last term. The reason for this is the lack of a constant term in the expansions for den[1] and den[3] above. Yet den[2] has a constant term 2. Does it have nonnegative coefficients in t.

In other words, we are at the end only interested in the second term involving B.

> **ansatz := A/(t-den[1]) + B/(t-den[2]) + C/(t-den[3]);**

$$\text{ansatz} := \frac{A}{t-x} + \frac{B}{t+x-\frac{1}{2}-\frac{\sqrt{-4x+1}}{2}} + \frac{C}{t+x-\frac{1}{2}+\frac{\sqrt{-4x+1}}{2}} \quad (1.3)$$

We solve for the unknowns A,B, and C.

```
> simplify(GT)*(2*x-1) = ansatz;
GTeq:=numer(GT) = (A*(t-den[2])*(t-den[3]) + B*(t-den[1])*(t-
den[3]) + C*(t-den[1])*(t-den[2]));
ABCsubs:= map(simplify,solve({seq(map(coeff,GTeq,t,i),i=0..2)},
{A,B,C}));
map(gdev,subs(ABCsubs,[A,B,C]),x=0,2);
```

$$-\frac{x(t^2+(4x-2)t+x^2)}{(t-x)(t^2+2xt+x^2-t)} = \frac{A}{t-x} + \frac{B}{t+x-\frac{1}{2}-\frac{\sqrt{-4x+1}}{2}}$$

$$+ \frac{C}{t+x-\frac{1}{2}+\frac{\sqrt{-4x+1}}{2}}$$

$$\text{ABCsubs} := \left\{ A = \frac{-6x^2+2x}{4x-1}, B = \frac{x(2x-1)(\sqrt{-4x+1}+1)}{8x-2}, C = \right. \\ \left. - \frac{(\sqrt{-4x+1}-1)x(2x-1)}{8x-2} \right\}$$

$$[-2x-2x^2+O(x^3), x+x^2+O(x^3), x^2+3x^3+O(x^4)] \quad (1.4)$$

Inserting t=0 into the part involving B (which is legitimate as there are only non-negative powers of t) gives the diagonal generating function.

Note that we need to add the earlier omitted factor 1/(2*x-1).

In order to simplify it we compute an algebraic equation for it which we then solve.

We have to choose the solution with a zero constant term.

```
> subs(ABCsubs,1/(2*x-1)*B/(-den[2]));
gdev(%,x,8);
algfuntoalgeq(%%,y(x));
solve(%,y);
```

$$-\frac{x(\sqrt{-4x+1}+1)}{(8x-2)\left(-x+\frac{1}{2}+\frac{\sqrt{-4x+1}}{2}\right)} \\ x+5x^2+22x^3+93x^4+386x^5+1586x^6+6476x^7+26333x^8+O(x^9) \\ (16x^2-8x+1)y^2+(4x-1)y+x \\ \frac{\sqrt{-4x+1}}{2}-\frac{1}{2}, \frac{-1}{2}-\frac{\sqrt{-4x+1}}{2}}{4x-1}, \frac{-1}{4x-1} \quad (1.5)$$

This is the final solution of the diagonal of y*G.

```
> GGx := (1/2)*(1/(1-4*x) - 1/sqrt(1-4*x));
```

gdev(GGx,x,6);

$$GGx := \frac{1}{2(-4x+1)} - \frac{1}{2\sqrt{-4x+1}}$$

$$x + 5x^2 + 22x^3 + 93x^4 + 386x^5 + 1586x^6 + O(x^7) \quad (1.6)$$

We can however directly extract the coefficients in a closed form due to

**> sum(binomial(2*n,n)*z^n,n=0..infinity);
series(sum(binomial(2*n,n)*z^n,n=0..10),z,6);
gdev((1-4*z)^(-1/2),z,6);**

$$\sum_{n=0}^{\infty} \binom{2n}{n} z^n$$

$$1 + 2z + 6z^2 + 20z^3 + 70z^4 + 252z^5 + O(z^6)$$

$$1 + 2z + 6z^2 + 20z^3 + 70z^4 + 252z^5 + O(z^6) \quad (1.7)$$

This gives

**> GN := n-> 1/2*(4^n-binomial(2*n,n));
seq(GN(n),n=1..10);**

$$GN := n \mapsto \frac{4^n}{2} - \frac{\binom{2n}{n}}{2}$$

$$1, 5, 22, 93, 386, 1586, 6476, 26333, 106762, 431910 \quad (1.8)$$

And the asymptotics is directly computed from the known asymptotics of the central binomial coefficient. Note that we can compute it to any chose order.

**> 4^n/2*(1-map(simplify,asympt(binomial(2*n,n)/4^n,n,4)))
assuming n::posint;**

$$\frac{4^n \left(1 - \frac{1}{\sqrt{\pi} \sqrt{n}} + \frac{1}{8\sqrt{\pi} n^{3/2}} - \frac{1}{128\sqrt{\pi} n^{5/2}} - O\left(\frac{1}{n^{7/2}}\right) \right)}{2} \quad (1.9)$$

Proof of Proposition 4.4 - The asymptotics of the second moment

> yzsubs := {y = 1 + l*s/sqrt(n), z = 1 + l*t/sqrt(n)};

$$yzsubs := \left\{ y = 1 + \frac{ls}{\sqrt{n}}, z = 1 + \frac{lt}{\sqrt{n}} \right\} \quad (2.1)$$

**> B0 := 1-x*y^2*z^2/(1-4*x*y*z)*(1 + 2*x*y/(1-2*x*y*(1+y*z)) + 2*x*z/(1-2*x*z*(1+y*z)));
B1 := 1/(1-2*x*z*(1+y*z))*(1 - x*(1+y*z)^2 - x*y*z/(1-2*x*y*(1+y*z)) - x*y*z);
B2 := 1/(1-2*x*y*(1+y*z))*(1 - x*(1+y*z)^2 - x*y*z/(1-2*x*z*(1+y*z)) - x*y*z);
B3 := 1-x/(1-4*x*y*z)*(1 + 2*x*y^2*z/(1-2*x*y*(1+y*z)) + 2*x*y*z^2/(1-2*x*z*(1+y*z)));**

$$\begin{aligned}
B0 &:= 1 - \frac{xy^2 z^2 \left(1 + \frac{2xy}{1-2xy(yz+1)} + \frac{2xz}{1-2xz(yz+1)}\right)}{-4xyz+1} \\
B1 &:= \frac{1-x(yz+1)^2 - \frac{xyz}{1-2xy(yz+1)} - xyz}{1-2xz(yz+1)} \\
B2 &:= \frac{1-x(yz+1)^2 - \frac{xyz}{1-2xz(yz+1)} - xyz}{1-2xy(yz+1)} \\
B3 &:= 1 - \frac{x \left(1 + \frac{2xy^2 z}{1-2xy(yz+1)} + \frac{2xyz^2}{1-2xz(yz+1)}\right)}{-4xyz+1} \tag{2.2}
\end{aligned}$$

We need $[x^n y^{(n-1)} z^{(n-1)}] F(x,y,z)$ which is equivalent to $[x^n y^n z^n] yzF(x,y,z)$ being the diagonal of a function.

> F := y*z*(B0+B1+B2+B3);

$$\begin{aligned}
F &:= yz \left(2 - \frac{xy^2 z^2 \left(1 + \frac{2xy}{1-2xy(yz+1)} + \frac{2xz}{1-2xz(yz+1)}\right)}{-4xyz+1} \right. \\
&\quad + \frac{1-x(yz+1)^2 - \frac{xyz}{1-2xy(yz+1)} - xyz}{1-2xz(yz+1)} \\
&\quad + \frac{1-x(yz+1)^2 - \frac{xyz}{1-2xz(yz+1)} - xyz}{1-2xy(yz+1)} \\
&\quad \left. - \frac{x \left(1 + \frac{2xy^2 z}{1-2xy(yz+1)} + \frac{2xyz^2}{1-2xz(yz+1)}\right)}{-4xyz+1} \right) \tag{2.3}
\end{aligned}$$

We use the notation H for the function D(x,y,z) in the paper as D is the operator for differentiation in Maple.

> H := 1 - x*(1+y*z)^2 - x*y*z/(1-2*x*z*(1+y*z)) - x*y*z/(1-2*x*y*(1+y*z));

$$H := 1 - x(yz+1)^2 - \frac{xyz}{1-2xz(yz+1)} - \frac{xyz}{1-2xy(yz+1)} \tag{2.4}$$

Expand at $(x,y,z) = (1/8,1,1)$

> solve(subs([y=1,z=1],H),x);

$$\frac{1}{8}, \frac{1}{2} \tag{2.5}$$

> #degree of expansion

N := 8:

#expansion point

```

x0 := 1/8:
y0 := 1:
z0 := 1:

A := 'A':
Y := 'Y': Z := 'Z':
n := 'n': k := 'k':

#expansion point of x
xB := x0:

#shift y and z such that Y=y-y0, Z=z-z0
#then expansion at Y=0,Z=0
Hs := subs([y=Y+y0,z=Z+z0],H):

#degree of expansion
for n from 1 to N do

#all possibilites for coefficients of Y=(y-1) and Z=(z-1)
for k from 0 to n do

#new ansatz
xB := xB + A*Y^k*Z^(n-k);

#substitute ansatz into implicit equation H
impSubs := subs(x=xB, Hs);

#distinguish cases (problems with diff if 0 times
differentiated)
if k = 0 then
  impDif := diff(impSubs,Z$(n-k));
else if n = k then
  impDif := diff(impSubs,Y$k);
else
  impDif := diff(diff(impSubs,Y$k),Z$(n-k));
end if; end if;

#solve for variable A
tmp := [solve(eval(impDif,[Y=0,Z=0])=0,A)];

if nops(tmp) > 1 then
  printf("ERROR: At ansatz for (y-1)^%d*(z-1)^%d %d
branches detected", k,n-k,nops(tmp));
  break;
end if;

XB[n,k] := op(1,tmp);
xB := subs(A=XB[n,k],xB);

```

```
end do;
end do;
```

```
n := 'n': k := 'k':
```

```
printf("This is f(Y,Z) with Y=(y-1) and Z=(z-1):");
sort(xB,[Z,Y],ascending);
```

This is f(Y,Z) with Y=(y-1) and Z=(z-1):

$$\begin{aligned}
& \frac{1}{8} - \frac{1}{8} Y - \frac{1}{8} Z + \frac{3}{32} Y^2 + \frac{1}{8} ZY + \frac{3}{32} Z^2 - \frac{1}{16} Y^3 - \frac{3}{32} ZY^2 - \frac{3}{32} Z^2 Y \\
& - \frac{1}{16} Z^3 + \frac{5}{128} Y^4 + \frac{1}{16} ZY^3 + \frac{13}{192} Z^2 Y^2 + \frac{1}{16} Z^3 Y + \frac{5}{128} Z^4 - \frac{3}{128} Y^5 \\
& - \frac{5}{128} ZY^4 - \frac{1}{24} Z^2 Y^3 - \frac{1}{24} Z^3 Y^2 - \frac{5}{128} Z^4 Y - \frac{3}{128} Z^5 + \frac{7}{512} Y^6 \\
& + \frac{3}{128} ZY^5 + \frac{121}{4608} Z^2 Y^4 + \frac{7}{576} Z^3 Y^3 + \frac{121}{4608} Z^4 Y^2 + \frac{3}{128} Z^5 Y \\
& + \frac{7}{512} Z^6 - \frac{1}{128} Y^7 - \frac{7}{512} ZY^6 - \frac{11}{512} Z^2 Y^5 + \frac{19}{2304} Z^3 Y^4 + \frac{19}{2304} Z^4 Y^3 \\
& - \frac{11}{512} Z^5 Y^2 - \frac{7}{512} Z^6 Y - \frac{1}{128} Z^7 + \frac{9}{2048} Y^8 + \frac{1}{128} ZY^7 + \frac{311}{13824} Z^2 Y^6 \\
& - \frac{37}{3456} Z^3 Y^5 - \frac{1255}{27648} Z^4 Y^4 - \frac{37}{3456} Z^5 Y^3 + \frac{311}{13824} Z^6 Y^2 + \frac{1}{128} Z^7 Y \\
& + \frac{9}{2048} Z^8
\end{aligned} \tag{2.6}$$

```
> logxB := sort(expand(convert(gdev(convert(gdev(log(xB),Y=0,
N+1),polynom),Z=0,N+1),polynom)),[Z,Y],ascending);
```

$$\begin{aligned}
\log xB := & -3 \ln(2) - Y - Z + \frac{Y^2}{4} + \frac{Z^2}{4} - \frac{Y^3}{12} - \frac{Z^3}{12} + \frac{Y^4}{32} - \frac{Z^2 Y^2}{48} + \frac{Z^4}{32} - \frac{Y^5}{80} \\
& + \frac{Z^2 Y^3}{48} + \frac{Z^3 Y^2}{48} - \frac{Z^5}{80} + \frac{Y^6}{192} + \frac{7 Z^2 Y^4}{576} - \frac{13 Z^3 Y^3}{144} + \frac{7 Z^4 Y^2}{576} + \frac{Z^6}{192} \\
& - \frac{Y^7}{448} - \frac{13 Z^2 Y^5}{288} + \frac{53 Z^3 Y^4}{576} + \frac{53 Z^4 Y^3}{576} - \frac{13 Z^5 Y^2}{288} - \frac{Z^7}{448} + \frac{Y^8}{1024} \\
& + \frac{419 Z^2 Y^6}{6912} - \frac{35 Z^3 Y^5}{864} - \frac{2239 Z^4 Y^4}{13824} - \frac{35 Z^5 Y^3}{864} + \frac{419 Z^6 Y^2}{6912} + \frac{Z^8}{1024} \\
& + \frac{9 ZY^8}{256} + \frac{947 Z^2 Y^7}{6912} - \frac{59 Z^3 Y^6}{6912} - \frac{271 Z^4 Y^5}{768} - \frac{271 Z^5 Y^4}{768} - \frac{59 Z^6 Y^3}{6912} \\
& + \frac{947 Z^7 Y^2}{6912} + \frac{9 Z^8 Y}{256} + \frac{277 Z^2 Y^8}{3456} + \frac{799 Z^3 Y^7}{6912} - \frac{667 Z^4 Y^6}{4608} - \frac{319 Z^5 Y^5}{864} \\
& - \frac{667 Z^6 Y^4}{4608} + \frac{799 Z^7 Y^3}{6912} + \frac{277 Z^8 Y^2}{3456} + \frac{1487 Z^3 Y^8}{27648} + \frac{161 Z^4 Y^7}{13824}
\end{aligned} \tag{2.7}$$

$$\begin{aligned}
& - \frac{1445 Z^5 Y^6}{13824} - \frac{1445 Z^6 Y^5}{13824} + \frac{161 Z^7 Y^4}{13824} + \frac{1487 Z^8 Y^3}{27648} + \frac{2959 Z^4 Y^8}{221184} \\
& - \frac{7 Z^5 Y^7}{1152} - \frac{629 Z^6 Y^6}{18432} - \frac{7 Z^7 Y^5}{1152} + \frac{2959 Z^8 Y^4}{221184} - \frac{1921 Z^5 Y^8}{331776} \\
& + \frac{763 Z^6 Y^7}{165888} + \frac{763 Z^7 Y^6}{165888} - \frac{1921 Z^8 Y^5}{331776} + \frac{8003 Z^6 Y^8}{3981312} - \frac{2837 Z^7 Y^7}{165888} \\
& + \frac{8003 Z^8 Y^6}{3981312} + \frac{2173 Z^7 Y^8}{3981312} + \frac{2173 Z^8 Y^7}{3981312} + \frac{3784235 Z^8 Y^8}{191102976}
\end{aligned}$$

Substitute the values of Y and Z

> fyz := subs([Y=y-y0,Z=z-z0],xB):

logfyz := subs([Y=y-y0,Z=z-z0],logxB):

**> logglambdayz := convert(gdev(convert(gdev(logfyz + log(y) + log(z), z=1,N+1),polynom),y=1,N+1),polynom);
subs(yzsubs,%):
exptmp := map(simplify@(x->-(n+1)*x),convert(gdev(%,n=infinity,N),polynom));**

$$\begin{aligned}
\text{logglambdayz} := & -3 \ln(2) - \frac{(1-z)^2}{4} - \frac{(1-z)^3}{4} - \frac{7(1-z)^4}{32} - \frac{3(1-z)^5}{16} \\
& - \frac{31(1-z)^6}{192} - \frac{9(1-z)^7}{64} - \frac{127(1-z)^8}{1024} - \frac{9(1-z)^8(1-y)}{256} + \left(-\frac{1}{4} \right. \\
& - \frac{(1-z)^2}{48} - \frac{(1-z)^3}{48} + \frac{7(1-z)^4}{576} + \frac{13(1-z)^5}{288} + \frac{419(1-z)^6}{6912} \\
& \left. - \frac{947(1-z)^7}{6912} + \frac{277(1-z)^8}{3456} \right) (1-y)^2 + \left(-\frac{1}{4} - \frac{(1-z)^2}{48} \right. \\
& - \frac{13(1-z)^3}{144} - \frac{53(1-z)^4}{576} - \frac{35(1-z)^5}{864} + \frac{59(1-z)^6}{6912} + \frac{799(1-z)^7}{6912} \\
& \left. - \frac{1487(1-z)^8}{27648} \right) (1-y)^3 + \left(-\frac{7}{32} + \frac{7(1-z)^2}{576} - \frac{53(1-z)^3}{576} \right. \\
& - \frac{2239(1-z)^4}{13824} + \frac{271(1-z)^5}{768} - \frac{667(1-z)^6}{4608} - \frac{161(1-z)^7}{13824} \\
& \left. + \frac{2959(1-z)^8}{221184} \right) (1-y)^4 + \left(-\frac{3}{16} + \frac{13(1-z)^2}{288} - \frac{35(1-z)^3}{864} \right. \\
& + \frac{271(1-z)^4}{768} - \frac{319(1-z)^5}{864} + \frac{1445(1-z)^6}{13824} - \frac{7(1-z)^7}{1152} \\
& \left. + \frac{1921(1-z)^8}{331776} \right) (1-y)^5 + \left(-\frac{31}{192} + \frac{419(1-z)^2}{6912} + \frac{59(1-z)^3}{6912} \right. \\
& \left. - \frac{667(1-z)^4}{4608} + \frac{1445(1-z)^5}{13824} - \frac{629(1-z)^6}{18432} - \frac{763(1-z)^7}{165888} \right)
\end{aligned}$$

$$\begin{aligned}
& + \frac{8003(1-z)^8}{3981312} \Big) (1-y)^6 + \left(-\frac{9}{64} - \frac{947(1-z)^2}{6912} + \frac{799(1-z)^3}{6912} \right. \\
& - \frac{161(1-z)^4}{13824} - \frac{7(1-z)^5}{1152} - \frac{763(1-z)^6}{165888} - \frac{2837(1-z)^7}{165888} \\
& \left. - \frac{2173(1-z)^8}{3981312} \Big) (1-y)^7 + \left(-\frac{163}{1024} + \frac{9z}{256} + \frac{277(1-z)^2}{3456} \right. \\
& - \frac{1487(1-z)^3}{27648} + \frac{2959(1-z)^4}{221184} + \frac{1921(1-z)^5}{331776} + \frac{8003(1-z)^6}{3981312} \\
& \left. - \frac{2173(1-z)^7}{3981312} + \frac{3784235(1-z)^8}{191102976} \Big) (1-y)^8
\end{aligned}$$

$$\begin{aligned}
\text{exptmp} := & 3(n+1)\ln(2) - \frac{(n+1)(s^2+t^2)}{4n} + \frac{I(n+1)(s^3+t^3)\sqrt{\frac{1}{n}}}{4n} & (2.8) \\
& + \frac{(n+1)(21s^4+2s^2t^2+21t^4)}{96n^2} \\
& - \frac{I(n+1)(9s^5+s^3t^2+s^2t^3+9t^5)\sqrt{\frac{1}{n}}}{48n^2} \\
& - \frac{(n+1)(93s^6-7s^4t^2+52s^3t^3-7s^2t^4+93t^6)}{576n^3} \\
& + \frac{I(n+1)(81s^7-26s^5t^2+53s^4t^3+53s^3t^4-26s^2t^5+81t^7)\sqrt{\frac{1}{n}}}{576n^3} \\
& + \frac{1}{27648n^4} ((n+1)(3429s^8-1676s^6t^2+1120s^5t^3+4478s^4t^4 \\
& + 1120s^3t^5-1676s^2t^6+3429t^8))
\end{aligned}$$

Extract the terms which are responsible for the exponential growth and the ones leading to a normal approximation.

```

> expgrowth := subs([s=0,t=0],convert(exptmp,polynomial));
#domterms := expgrowth-(1/4)*t^2-(1/4)*s^2-I*c*t-I*c*s;
domterms := expgrowth-(1/4)*t^2-(1/4)*s^2;
expgrowth := 3(n+1)ln(2)

```

$$\text{domterms} := 3(n+1)\ln(2) - \frac{t^2}{4} - \frac{s^2}{4} \quad (2.9)$$

Compute the other factor of the integrand: $-F/(y*z*H_x)$ without $y*z$ in denominator! (extract $[x^{(n-1)} y^{(n-1)} z^{(n-1)}]$)

```

> mtaylor(subs([y=Y+1,z=Z+1],subs(x=fyz,-F/(diff(H,x)))), [Y=0,Z=0], N+1):

```

```

sort(%, [Z, Y], ascending);
expand(subs(yzsubs, subs([Y=y-1, Z=z-1], %))):
sort(%, n);
Pnst := convert(gdev(%, n=infinity, N-2), polynom);

```

$$\begin{aligned}
& \frac{1}{8} - \frac{1}{32} Y^2 - \frac{1}{32} Z^2 + \frac{1}{32} Y^3 + \frac{1}{32} Z^3 - \frac{3}{128} Y^4 + \frac{1}{192} Z^2 Y^2 - \frac{3}{128} Z^4 \\
& + \frac{1}{64} Y^5 - \frac{1}{192} Z^2 Y^3 - \frac{1}{192} Z^3 Y^2 + \frac{1}{64} Z^5 - \frac{5}{512} Y^6 + \frac{37}{4608} Z^2 Y^4 \\
& - \frac{1}{288} Z^3 Y^3 + \frac{37}{4608} Z^4 Y^2 - \frac{5}{512} Z^6 + \frac{3}{512} Y^7 - \frac{25}{2304} Z^2 Y^5 \\
& + \frac{23}{4608} Z^3 Y^4 + \frac{23}{4608} Z^4 Y^3 - \frac{25}{2304} Z^5 Y^2 + \frac{3}{512} Z^7 - \frac{7}{2048} Y^8 \\
& + \frac{149}{13824} Z^2 Y^6 + \frac{1}{432} Z^3 Y^5 - \frac{463}{27648} Z^4 Y^4 + \frac{1}{432} Z^5 Y^3 + \frac{149}{13824} Z^6 Y^2 \\
& - \frac{7}{2048} Z^8
\end{aligned}$$

$$\begin{aligned}
& \frac{t^2}{32n} + \frac{s^2}{32n} - \frac{1s^3}{32n^{3/2}} - \frac{1t^3}{32n^{3/2}} - \frac{3t^4}{128n^2} - \frac{3s^4}{128n^2} + \frac{t^2s^2}{192n^2} + \frac{1s^5}{64n^{5/2}} \\
& + \frac{1t^5}{64n^{5/2}} - \frac{1t^3s^2}{192n^{5/2}} - \frac{1t^2s^3}{192n^{5/2}} - \frac{37t^4s^2}{4608n^3} + \frac{t^3s^3}{288n^3} - \frac{37t^2s^4}{4608n^3} \\
& + \frac{5t^6}{512n^3} + \frac{5s^6}{512n^3} - \frac{31t^7}{512n^{7/2}} - \frac{31s^7}{512n^{7/2}} + \frac{251t^2s^5}{2304n^{7/2}} - \frac{231t^3s^4}{4608n^{7/2}} \\
& - \frac{231t^4s^3}{4608n^{7/2}} + \frac{251t^5s^2}{2304n^{7/2}} - \frac{7t^8}{2048n^4} - \frac{7s^8}{2048n^4} + \frac{149t^2s^6}{13824n^4} + \frac{t^3s^5}{432n^4} \\
& - \frac{463t^4s^4}{27648n^4} + \frac{t^5s^3}{432n^4} + \frac{149t^6s^2}{13824n^4} + \frac{1}{8}
\end{aligned}$$

$$\begin{aligned}
Pnst := & \frac{1}{8} + \frac{\frac{t^2}{32} + \frac{s^2}{32}}{n} + \left(-\frac{1s^3}{32} - \frac{1t^3}{32} \right) \left(\frac{1}{n} \right)^{3/2} \\
& + \frac{-\frac{3}{128} t^4 + \frac{1}{192} s^2 t^2 - \frac{3}{128} s^4}{n^2} + \left(-\frac{1t^3 s^2}{192} - \frac{1t^2 s^3}{192} + \frac{1s^5}{64} \right. \\
& \left. + \frac{1t^5}{64} \right) \left(\frac{1}{n} \right)^{5/2} \\
& + \frac{\frac{5}{512} t^6 - \frac{37}{4608} s^4 t^2 + \frac{1}{288} s^3 t^3 - \frac{37}{4608} s^2 t^4 + \frac{5}{512} s^6}{n^3}
\end{aligned} \tag{2.10}$$

Expansion in n, careful: square-roots in n
> NORD := N:

XX := exptmp-domterms:

map(simplify,gdev(Pnst*(add(XX^k/factorial(k),k=0..NORD)),n=

infinity,NORD-1)) assuming n::posint;

intfact := convert(%,polynom):

$$\begin{aligned}
 & \frac{1}{8} + \frac{I(s^3 + t^3)}{32\sqrt{n}} + \frac{-3s^6 - 6s^3t^3 - 3t^6 + 21s^4 + 2s^2t^2 + 21t^4}{768n} & (2.11) \\
 & - \frac{1}{3072n^{3/2}} (I(s^9 + 3s^6t^3 + 3s^3t^6 + t^9 - 21s^7 - 2s^5t^2 - 21s^4t^3 - 21s^3t^4 \\
 & - 2s^2t^5 - 21t^7 + 72s^5 + 8s^3t^2 + 8s^2t^3 + 72t^5)) + \frac{1}{147456n^2} (3s^{12} \\
 & - 126s^{10} + 12s^9t^3 + (-12t^2 + 1305)s^8 - 252s^7t^3 + (18t^6 - 126t^4 \\
 & + 180t^2 - 2976)s^6 + (-24t^5 + 960t^3)s^5 + (-126t^6 + 886t^4 + 224t^2)s^4 \\
 & + (12t^9 - 252t^7 + 960t^5 - 1664t^3)s^3 + (-12t^8 + 180t^6 + 224t^4)s^2 \\
 & + 3t^6(t^6 - 42t^4 + 435t^2 - 992)) + \frac{1}{2949120n^{5/2}} (3Is^{15} - 210Is^{13} \\
 & + 15It^3s^{12} + (4365I - 20It^2)s^{11} - 630It^3s^{10} + I(30t^6 - 210t^4 + 660t^2 \\
 & - 30000)s^9 - 60I\left(t^2 - \frac{451}{4}\right)t^3s^8 + I(-630t^6 + 4430t^4 - 2000t^2 \\
 & + 51840)s^7 + 30It^3\left(t^6 - 21t^4 + 102t^2 - \frac{2488}{3}\right)s^6 - 60It^2\left(t^6 - 51t^4 \\
 & + 236t^2 + \frac{832}{3}\right)s^5 + I(-210t^9 + 4430t^7 - 14160t^5 + 33920t^3)s^4 \\
 & + 15It^4\left(t^8 - 42t^6 + 451t^4 - \frac{4976}{3}t^2 + \frac{6784}{3}\right)s^3 - 20It^5(t^6 - 33t^4 \\
 & + 100t^2 + 832)s^2 + 3It^7(t^8 - 70t^6 + 1455t^4 - 10000t^2 + 17280) \\
 & + \frac{1}{212336640n^3} (-9s^{18} + 945s^{16} - 54s^{15}t^3 + (90t^2 - 32805)s^{14} \\
 & + 3780s^{13}t^3 + (-135t^6 + 945t^4 - 5220t^2 + 452385)s^{12} + (360t^5 \\
 & - 80010t^3)s^{11} + (5670t^6 - 39870t^4 + 59310t^2 - 2337120)s^{10} \\
 & - 180t^3(t^6 - 21t^4 + 138t^2 - 3584)s^9 + (540t^8 - 66825t^6 + 405135t^4 \\
 & + 177120t^2 + 3291840)s^8 + (3780t^9 - 79740t^7 + 311040t^5 \\
 & - 2171520t^3)s^7 + (-135t^{12} + 5670t^{10} - 66825t^8 + 504580t^6
 \end{aligned}$$

$$\begin{aligned}
& -1476480 t^4 - 1608960 t^2) s^6 + 360 t^3 \left(t^8 - 69 t^6 + 864 t^4 - \frac{3296}{3} t^2 \right. \\
& \left. + \frac{8960}{3} \right) s^5 + (945 t^{12} - 39870 t^{10} + 405135 t^8 - 1476480 t^6 \\
& + 4298880 t^4) s^4 + (-54 t^{15} + 3780 t^{13} - 80010 t^{11} + 645120 t^9 \\
& - 2171520 t^7 + 1075200 t^5) s^3 + 90 t^6 \left(t^8 - 58 t^6 + 659 t^4 + 1968 t^2 \right. \\
& \left. - \frac{53632}{3} \right) s^2 - 9 t^8 (t^{10} - 105 t^8 + 3645 t^6 - 50265 t^4 + 259680 t^2 \\
& - 365760) \Big) + O\left(\frac{1}{n^{7/2}}\right)
\end{aligned}$$

> terms := [op(intfact)]:

The generic result of the needed integrals. The value just depends on the needed power (i.e., moment).

See Lemma 4.7.

> intval := proc(k::integer)

local ret;

if k mod 2 = 1 then

ret := 0;

else

ret := sqrt(Pi)*2*factorial(k)/factorial(k/2);

end if;

return ret;

end proc:

There is a shift in k needed as we divide by k (or s, respectively)

> intvalshift := proc(k)

if k=0 then

return -Pi*I;

else if k > 0 then

return intval(k-1);

else

printf("ERROR: negative k in intval!");

end if;

end if;

end proc:

This is the term-wise computation of the integral.

> subexpasy := 0:

for tmp in terms do

tmp;

#TODO: not optimal, depends on structure of terms,

#better: extract coefficients with respect to n!!!

if not has(tmp,n) then

```

#case without n
prefac := 1;
ordn := 1;
stpart := tmp;
else
prefac := op(1,tmp);
ordn := op(3,tmp);
stpart := expand(op(2,tmp));
end if;

if not evalb(whattype(stpart)='+') then
liststpart := [stpart];
else
liststpart := [op(stpart)];
end if;
liststpart;

if nops(tmp) > 3 then
printf("ERROR: tmp has more then 3 terms!");
break
end if;

cur := 0;
for tt in liststpart do
sdeg := degree(tt,s);
tdeg := degree(tt,t);

cur := cur + intvalshift(sdeg)*intvalshift(tdeg)*
tt/s^sdeg/t^tdeg;
end do;
sdeg,tdeg,cur,prefac,ordn;
cur := simplify(cur*prefac*ordn);
subexpasy := subexpasy + cur;
end do;
subexpasy := map(simplify@(x->x/(2*Pi*I)^2),subexpasy) assuming
n::posint;

```

$$\begin{aligned}
subexpasy := & \frac{1}{32} - \frac{1}{16\sqrt{\pi}\sqrt{n}} + \frac{1}{32\pi n} + \frac{1}{128\sqrt{\pi}n^{3/2}} + \frac{17}{576\pi n^2} \\
& - \frac{1}{2048\sqrt{\pi}n^{5/2}} + \frac{481}{2304\pi n^3}
\end{aligned} \tag{2.12}$$

The main result: The asymptotics of $[x^{(n)} y^{(n-1)} z^{(n-1)}]F(x,y,z)$

```

> asyn := simplify(exp(expgrowth))*(subexpasy) assuming
n::posint;

```

$$asyn := 2^{3n+3} \left(\frac{1}{32} - \frac{1}{16\sqrt{\pi}\sqrt{n}} + \frac{1}{32\pi n} + \frac{1}{128\sqrt{\pi}n^{3/2}} + \frac{17}{576\pi n^2} \right) \tag{2.13}$$

$$-\frac{1}{2048\sqrt{\pi}n^{5/2}} + \frac{481}{2304\pi n^3}$$

and a simplified expression

> simplify(exp(expgrowth)/8)*(map(x->8*x,subexpasy)) assuming n::posint;

$$8^n \left(\frac{1}{4} - \frac{1}{2\sqrt{\pi}\sqrt{n}} + \frac{1}{4\pi n} + \frac{1}{16\sqrt{\pi}n^{3/2}} + \frac{17}{72\pi n^2} - \frac{1}{256\sqrt{\pi}n^{5/2}} + \frac{481}{288\pi n^3} \right) \quad (2.14)$$